NAG Library Routine Document

F07ADF (DGETRF)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F07ADF (DGETRF) computes the LU factorization of a real m by n matrix.

2 Specification

```
SUBROUTINE F07ADF (M, N, A, LDA, IPIV, INFO)
INTEGER            M, N, LDA, IPIV(min(M,N)), INFO
REAL (KIND=nag_wp) A(LDA,*)
```

The routine may be called by its LAPACK name ***dgetrf***.

3 Description

F07ADF (DGETRF) forms the LU factorization of a real m by n matrix A as $A = PLU$, where P is a permutation matrix, L is lower triangular with unit diagonal elements (lower trapezoidal if $m > n$) and U is upper triangular (upper trapezoidal if $m < n$). Usually A is square $(m = n)$, and both L and U are triangular. The routine uses partial pivoting, with row interchanges.

4 References

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Arguments

1: M – INTEGER *Input*

On entry: m, the number of rows of the matrix A.

Constraint: $\mathrm{M} \geq 0$.

2: N – INTEGER *Input*

On entry: n, the number of columns of the matrix A.

Constraint: $\mathrm{N} \geq 0$.

3: $\mathrm{A}(\mathrm{LDA}, *)$ – REAL (KIND=nag_wp) array *Input/Output*

Note: the second dimension of the array A must be at least $\max(1, \mathrm{N})$.

On entry: the m by n matrix A.

On exit: the factors L and U from the factorization $A = PLU$; the unit diagonal elements of L are not stored.

4: LDA – INTEGER *Input*

On entry: the first dimension of the array A as declared in the (sub)program from which F07ADF (DGETRF) is called.

Constraint: $\mathrm{LDA} \geq \max(1, \mathrm{M})$.

5: IPIV(min(M, N)) – INTEGER array *Output*

On exit: the pivot indices that define the permutation matrix. At the ith step, if $\mathrm{IPIV}(i) > i$ then row i of the matrix A was interchanged with row $\mathrm{IPIV}(i)$, for $i = 1, 2, \ldots, \min(m, n)$. $\mathrm{IPIV}(i) \leq i$ indicates that, at the ith step, a row interchange was not required.

6: INFO – INTEGER *Output*

On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO $= -i$, argument i had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

Element ⟨*value*⟩ of the diagonal is exactly zero. The factorization has been completed, but the factor U is exactly singular, and division by zero will occur if it is used to solve a system of equations.

7 Accuracy

The computed factors L and U are the exact factors of a perturbed matrix $A + E$, where

$$|E| \leq c(\min(m, n))\epsilon P|L||U|,$$

$c(n)$ is a modest linear function of n, and ϵ is the ***machine precision***.

8 Parallelism and Performance

F07ADF (DGETRF) is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

F07ADF (DGETRF) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

The total number of floating-point operations is approximately $\frac{2}{3}n^3$ if $m = n$ (the usual case), $\frac{1}{3}n^2(3m - n)$ if $m > n$ and $\frac{1}{3}m^2(3n - m)$ if $m < n$.

A call to this routine with $m = n$ may be followed by calls to the routines:

F07AEF (DGETRS) to solve $AX = B$ or $A^{\mathrm{T}}X = B$;

F07AGF (DGECON) to estimate the condition number of A;

F07AJF (DGETRI) to compute the inverse of A.

The complex analogue of this routine is F07ARF (ZGETRF).

10 Example

This example computes the LU factorization of the matrix A, where

$$A = \begin{pmatrix} 1.80 & 2.88 & 2.05 & -0.89 \\ 5.25 & -2.95 & -0.95 & -3.80 \\ 1.58 & -2.69 & -2.90 & -1.04 \\ -1.11 & -0.66 & -0.59 & 0.80 \end{pmatrix}.$$

10.1 Program Text

```
    Program f07adfe

!     F07ADF Example Program Text

!     Mark 26 Release. NAG Copyright 2016.

!     .. Use Statements ..
      Use nag_library, Only: dgetrf, nag_wp, x04caf
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: nin = 5, nout = 6
!     .. Local Scalars ..
      Integer                          :: i, ifail, info, lda, m, n
!     .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable  :: a(:,:)
      Integer, Allocatable             :: ipiv(:)
!     .. Intrinsic Procedures ..
      Intrinsic                        :: min
!     .. Executable Statements ..
      Write (nout,*) 'F07ADF Example Program Results'
!     Skip heading in data file
      Read (nin,*)
      Read (nin,*) m, n
      lda = m
      Allocate (a(lda,n),ipiv(n))

!     Read A from data file

      Read (nin,*)(a(i,1:n),i=1,m)

!     Factorize A

!     The NAG name equivalent of dgetrf is f07adf
      Call dgetrf(m,n,a,lda,ipiv,info)

!     Print details of factorization

      Write (nout,*)
      Flush (nout)

!     ifail: behaviour on error exit
!             =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
      ifail = 0
      Call x04caf('General',' ',m,n,a,lda,'Details of factorization',ifail)

!     Print pivot indices

      Write (nout,*)
      Write (nout,*) 'IPIV'
      Write (nout,99999) ipiv(1:min(m,n))

      If (info/=0) Then
        Write (nout,*) 'The factor U is singular'
      End If

99999 Format ((3X,7I11))
    End Program f07adfe
```

10.2 Program Data

```
F07ADF Example Program Data
  4  4                              :Values of M and N
  1.80   2.88   2.05  -0.89
  5.25  -2.95  -0.95  -3.80
  1.58  -2.69  -2.90  -1.04
 -1.11  -0.66  -0.59   0.80   :End of matrix A
```

10.3 Program Results

```
 F07ADF Example Program Results

 Details of factorization
             1           2           3           4
 1      5.2500     -2.9500     -0.9500     -3.8000
 2      0.3429      3.8914      2.3757      0.4129
 3      0.3010     -0.4631     -1.5139      0.2948
 4     -0.2114     -0.3299      0.0047      0.1314

 IPIV
             2           2           3           4
```